Automatic expansion of C-fractions for solutions of Riccati equations via Guessing and Proving.

The examples from Cuyt *et alii*., Handbook of Continued Fractions for Special Functions.

Maulat & Salvy

This worksheet demonstrates the use of the submodule gfun:-ContFrac, for power series solutions of Riccati differential equations : y'=p y^2 + q y + r with p,q,r rational in the variable z.

The first section gives a few examples, and demonstrates how to print some information along the way.
A comprehensive list of all the examples from Cuyt *et alii* is then given in the following section.

```
> restart;
> starttime := time();
```

$$starttime := 0.102 \quad (1)$$

```
> with(gfun):

  if gfun:-version() < 3.70 then
    error "Old gfun version. Please download the latest version on gfun's website, http://perso.ens-lyon.fr/bruno.salvy/software/the-gfun-package/."
  fi;

  with(gfun:-ContFrac):
```

First examples.

The procedure Riccati_to_Cfrac takes as input :
- f: a function to be expanded (or a series)
- y: a function name
- z: the variable name

and optionally :
- a number (the series expansion order)

```
> Riccati_to_Cfrac( exp(z), y, z );
```

(1.1)

$$\left\{\frac{d}{dz}\,y(z) - y(z),\ y(0) = 1\right\}, \text{"implies"},\ y(z) = 1 + \cfrac{z}{1 + \cfrac{a_1 z}{1 + \cfrac{\dots}{1 + \cfrac{a_n z}{1 + \dots}}}},\ a_n = \begin{cases} \dfrac{1}{2\,(n+1)} & n{::}even \\ -\dfrac{1}{2\,n} & n{::}odd \end{cases} \quad (1.1)$$

The infolevel command enables to print details on the computation.
The 'demo' field concerns the main computation lines ; it can be set to 0 or 1.

```
> infolevel[demo]:=1;
```

$$infolevel_{demo} := 1 \quad (1.2)$$

```
> Riccati_to_Cfrac( tan(z), y, z, 20 );
computing series...
... done.
computing a Riccati equation (using a guessing approach)...
... done.
computing a C-fraction expansion...
... done. (in .669 seconds)
```

$$\left\{\frac{d}{dz}\,y(z) - 1 - y(z)^2,\ y(0) = 0\right\}, \text{"implies"},\ y(z) = \cfrac{z}{1 + \cfrac{a_1 z^2}{1 + \cfrac{\dots}{1 + \cfrac{a_n z^2}{1 + \dots}}}},\ a_n = -\frac{1}{(2\,n-1)\,(2\,n+1)} \quad (1.3)$$

```
> infolevel[demo]:=0;
```

$$infolevel_{demo} := 0 \quad (1.4)$$

More information on how the continued fraction expansion is computed and proved can be printed using the 'gfuncontfrac' information field.

```
> infolevel[gfuncontfrac]:=1;
```

$$infolevel_{gfuncontfrac} := 1 \quad (1.5)$$

```
> Riccati_to_Cfrac( tan(z), y, z, 20 );
Guessing a formula
conjecture (on 8 coefficients)
```

$$y(z) = \cfrac{z}{1 + \cfrac{a_1 z^2}{1 + \cfrac{...}{1 + \cfrac{a_n z^2}{1 + ...}}}}, \quad a_n = -\frac{1}{(2n-1)(2n+1)}$$

```
defining a sequence H(n,z), which relates to convergence.
lemma: the conjecture holds iff there exists an unbounded i
(n) s.t., H(i(n),z) tends to 0.
  (i.e., their valuations tend to infinity)
proving Limit(val(H(n,z)),n = infinity) = infinity:
- computing a recurrence for H(n,z).
  (which does not conclude)
```

$$... z^8 H(n) + ... z^4 H(n+1) + (... z^4 + ... z^2) H(n+2) + ... H(n+3) + ... H(n+4) = 0$$

```
- reducing the recurrence order for H(n,z)...
- ... done.
```

$$\{-z^2 H(n) + (4n^2 + 12n + 9) H(n+1), H(0) = -z^2\}$$

QED.

$$\left\{\frac{d}{dz} y(z) - 1 - y(z)^2, y(0) = 0\right\}, \text{"implies"}, y(z) = \cfrac{z}{1 + \cfrac{a_1 z^2}{1 + \cfrac{...}{1 + \cfrac{a_n z^2}{1 + ...}}}}, a_n = -\frac{1}{(2n-1)(2n+1)} \qquad \textbf{(1.6)}$$

```
> infolevel[demo]:=0;
  infolevel[gfuncontfrac]:=0;
```

$$infolevel_{demo} := 0$$

$$infolevel_{gfuncontfrac} := 0 \qquad \textbf{(1.7)}$$

C-fractions formulas of Riccati solutions in *Cuyt et alii*.

In this section, formulas from the compendium by Cuyt *et alii* (2008) are recovered automatically. All explicit C-fractions solutions of Riccati equations are recovered.

Formulas are stored in a table gfun:-ContFrac:-Cuyt_Cfrac, which is accessed as follows:

```
> chapter:=11;
```

$$chapter := 11 \qquad \textbf{(2.1)}$$

```
> labels(chapter);
```

$$[1.2, 1.3, 2.2, 2.4, 3.7, 4.5, 4.8, 5.5, 6.4, 6.9, 7.1, 7.2, 7.4] \quad \textbf{(2.2)}$$

```
> Cuyt_Cfrac[chapter][1.3];
```

$$e^z \quad \textbf{(2.3)}$$

In what follows, only the parameters related to the series exponents are given example values.

Chapter 11

```
> chapter := 11;
```

$$chapter := 11 \quad \textbf{(2.1.1)}$$

```
> labels(chapter);
```

$$[1.2, 1.3, 2.2, 2.4, 3.7, 4.5, 4.8, 5.5, 6.4, 6.9, 7.1, 7.2, 7.4] \quad \textbf{(2.1.2)}$$

```
> N:=20;
```

$$N := 20 \quad \textbf{(2.1.3)}$$

Generic expansion, with finite series as input.
The order of the series is doubled when the C-fraction coefficients are in z^2.

```
> for formula in labels(chapter) do

    print( sprintf("%a.%a",chapter,formula) );
    f := Cuyt_Cfrac[chapter][formula]:

    if formula in {1.3, 2.2, 7.1, 7.2} then
      S := MultiSeries:-series(f,z,N):
    elif formula = 7.4 then
      S := series(f,z,2*N);
    else
      S := MultiSeries:-series(f,z,2*N)
    fi;

    Riccati_to_Cfrac( S, y, z );

  od;
```

"11.1.2"

$$f := \frac{1}{6}\,\frac{e^z z^2 - 6\,e^z z - z^2 + 12\,e^z - 6\,z - 12}{e^z z - 2\,e^z + z + 2}$$

$$\left\{\frac{d}{dz}\,y(z) - \frac{1}{12}\,\frac{z^2 - 36\,y(z) - 36\,y(z)^2}{z},\ y(0) = 0\right\}, \text{"implies"}, y(z)$$

$$= \cfrac{a_0\,z^2}{1 + \cfrac{\ldots}{1 + \cfrac{a_n\,z^2}{1 + \ldots}}},\ a_n = \frac{1}{4\,(2\,n+3)\,(2\,n+5)}$$

"11.1.3"

$$f := e^{z}$$

$$\left\{\frac{d}{dz}\, y(z) - y(z),\ y(0) = 1\right\}, \text{"implies"}, y(z) = 1 + \cfrac{z}{1 + \cfrac{a_1 z}{1 + \cfrac{\ldots}{1 + \cfrac{a_n z}{1 + \ldots}}}}, a_n$$

$$= \begin{cases} \frac{1}{2\,(n+1)} & n::even \\ -\frac{1}{2\,n} & n::odd \end{cases}$$

"11.2.2"

$$f := -\frac{\ln(1+z) - z}{\ln(1+z)}$$

$$\left\{\frac{d}{dz}\, y(z) - \frac{z - (-z+1)\, y(z) - y(z)^2}{z^2 + z},\ y(0) = 0\right\}, \text{"implies"}, y(z)$$

$$= \cfrac{a_0 z}{1 + \cfrac{\ldots}{1 + \cfrac{a_n z}{1 + \ldots}}}, a_n = \begin{cases} \frac{1}{4}\,\frac{n+2}{n+1} & n::even \\ \frac{1}{4}\,\frac{n+1}{n+2} & n::odd \end{cases}$$

"11.2.4"

$$f := -\frac{\ln\left(-\frac{1+z}{z-1}\right) - 2\,z}{\ln\left(-\frac{1+z}{z-1}\right)}$$

$$\left\{\frac{d}{dz}\, y(z) - \frac{-z^2 - (z^2+1)\, y(z) - y(z)^2}{-z^3 + z},\ y(0) = 0\right\}, \text{"implies"}, y(z)$$

$$= \cfrac{a_0 z^2}{1 + \cfrac{\ldots}{1 + \cfrac{a_n z^2}{1 + \ldots}}}, a_n = -\frac{(n+1)^2}{(2\,n+1)\,(2\,n+3)}$$

"11.3.7"

$$f := -\frac{\tan(z) - z}{\tan(z)}$$

$$\left\{\frac{d}{dz}\, y(z) - \frac{-z^2 - y(z) - y(z)^2}{z},\ y(0) = 0\right\}, \text{"implies"}, y(z)$$

$$= \frac{a_0 z^2}{1 + \frac{\ldots}{1 + \frac{a_n z^2}{1 + \ldots}}}, \; a_n = -\frac{1}{(2\,n+1)\,(2\,n+3)}$$

"11.4.5"

$$f := -\frac{-z\sqrt{-z^2+1} + \arcsin(z)}{\arcsin(z)}$$

$$\left\{\frac{d}{dz}\,y(z) - \frac{-2\,z^2 - (2\,z^2+1)\,y(z) - y(z)^2}{-z^3+z}, \; y(0) = 0\right\}, \text{"implies"}, \; y(z)$$

$$= \frac{a_0 z^2}{1 + \frac{\ldots}{1 + \frac{a_n z^2}{1 + \ldots}}}, \; a_n = \begin{cases} -\dfrac{(n+1)\,(n+2)}{(2\,n+3)\,(2\,n+1)} & n{::}even \\ -\dfrac{(n+1)\,n}{(2\,n+3)\,(2\,n+1)} & n{::}odd \end{cases}$$

"11.4.8"

$$f := -\frac{\arctan(z) - z}{\arctan(z)}$$

$$\left\{\frac{d}{dz}\,y(z) - \frac{z^2 - (-z^2+1)\,y(z) - y(z)^2}{z^3+z}, \; y(0) = 0\right\}, \text{"implies"}, \; y(z)$$

$$= \frac{a_0 z^2}{1 + \frac{\ldots}{1 + \frac{a_n z^2}{1 + \ldots}}}, \; a_n = \frac{(n+1)^2}{(2\,n+1)\,(2\,n+3)}$$

"11.5.5"

$$f := -\frac{\tanh(z) - z}{\tanh(z)}$$

$$\left\{\frac{d}{dz}\,y(z) - \frac{z^2 - y(z) - y(z)^2}{z}, \; y(0) = 0\right\}, \text{"implies"}, \; y(z)$$

$$= \frac{a_0 z^2}{1 + \frac{\ldots}{1 + \frac{a_n z^2}{1 + \ldots}}}, \; a_n = \frac{1}{(2\,n+1)\,(2\,n+3)}$$

"11.6.4"

$$f := -\frac{-z\sqrt{z^2+1} + \operatorname{arcsinh}(z)}{\operatorname{arcsinh}(z)}$$

$$\left\{\frac{d}{dz}\,y(z)-\frac{2\,z^2-(-2\,z^2+1)\,y(z)-y(z)^2}{z^3+z},\ y(0)=0\right\},\ \text{"implies"},\ y(z)$$

$$=\frac{a_0\,z^2}{1+\frac{\dots}{1+\frac{a_n\,z^2}{1+\dots}}},\ a_n=\begin{cases}\frac{(n+1)\,(n+2)}{(2\,n+3)\,(2\,n+1)} & n::even\\ \frac{(n+1)\,n}{(2\,n+3)\,(2\,n+1)} & n::odd\end{cases}$$

"11.6.9"

$$f:=\operatorname{arctanh}(z)$$

$$\left\{\frac{d}{dz}\,y(z)+\frac{1}{z^2-1},\ y(0)=0\right\},\ \text{"implies"},\ y(z)=\frac{z}{1+\frac{a_1\,z^2}{1+\frac{\dots}{1+\frac{a_n\,z^2}{1+\dots}}}},\ a_n=$$

$$-\frac{n^2}{(2\,n-1)\,(2\,n+1)}$$

"11.7.1"

$$f:=-\frac{-\alpha\,z+(1+z)^{\alpha}-1}{(1+z)^{\alpha}-1}$$

$$\left\{\frac{d}{dz}\,y(z)-\frac{-\alpha\,z+z-(\alpha\,z-z+1)\,y(z)-y(z)^2}{z\,(1+z)},\ y(0)=0\right\},\ \text{"implies"},\ y(z)$$

$$=\frac{a_0\,z}{1+\frac{\dots}{1+\frac{a_n\,z}{1+\dots}}},\ a_n=\begin{cases}-\frac{1}{4}\,\frac{-n+2\,\alpha-2}{n+1} & n::even\\ \frac{1}{4}\,\frac{n+2\,\alpha+1}{n+2} & n::odd\end{cases}$$

"11.7.2"

$$f:=\frac{(1+z)^{\alpha}\,\alpha\,z-(1+z)^{\alpha}+1}{(1+z)^{\alpha}-1}$$

$$\left\{\frac{d}{dz}\,y(z)-\frac{\alpha\,z+z-(-\alpha\,z-z+1)\,y(z)-y(z)^2}{(1+z)\,z},\ y(0)=0\right\},\ \text{"implies"},\ y(z)$$

$$=\frac{a_0\,z}{1+\frac{\dots}{1+\frac{a_n\,z}{1+\dots}}},\ a_n=\begin{cases}\frac{1}{4}\,\frac{n+2\,\alpha+2}{n+1} & n::even\\ -\frac{1}{4}\,\frac{-n+2\,\alpha-1}{n+2} & n::odd\end{cases}$$

"11.7.4"

$$f := \frac{\left(\left(\frac{1+\frac{1}{z}}{\frac{1}{z}-1}\right)^{\alpha}\alpha - \frac{\left(\frac{1+\frac{1}{z}}{\frac{1}{z}-1}\right)^{\alpha}}{z} + \alpha + \frac{1}{z}\right) z}{\left(\frac{1+\frac{1}{z}}{\frac{1}{z}-1}\right)^{\alpha} - 1}$$

$$\left\{\frac{d}{dz}\, y(z) + \frac{\alpha^2 z^2 - z^2 - (z^2+1)\, y(z) - y(z)^2}{z\,(z^2-1)},\ y(0) = 0\right\}, \text{"implies"}, y(z) = \frac{a_0 z^2}{1 + \frac{\dots}{1 + \frac{a_n z^2}{1+\dots}}},\ a_n = \frac{(\alpha+1+n)\,(\alpha-1-n)}{(2\,n+3)\,(2\,n+1)} \quad \textbf{(2.1.4)}$$

```
> f := -(-2*z+(1+z)^2-1)/((1+z)^2-1);
```

$$f := -\frac{-2\,z + (1+z)^2 - 1}{(1+z)^2 - 1} \quad \textbf{(2.1.5)}$$

```
> factor(f);
```

$$-\frac{z}{z+2} \quad \textbf{(2.1.6)}$$

Chapter 12

```
> chapter := 12;
```

$$chapter := 12 \quad \textbf{(2.2.1)}$$

```
> N:=25:
```

The parameter l in formulas (12.6.21) and (12.6.22) is set to a consistent value.

```
> vall:=3;
```

$$vall := 3 \quad \textbf{(2.2.2)}$$

```
> infolevel[gfuncontfrac]:=0;
```

$$infolevel_{gfuncontfrac} := 0 \quad \textbf{(2.2.3)}$$

```
> for formula in labels(chapter) do

    print( sprintf("%a.%a",chapter,formula) );
    f := Cuyt_Cfrac[chapter][formula]:

    if formula in {6.21,6.22} then
```

```
      print( f, l=vall );
      f := subs( l=vall, f );
    fi;

    Riccati_to_Cfrac( MultiSeries:-series(f,z,N), y, z )

  od:
```

"12.6.17"

$$\left\{\frac{d}{dz}y(z)+\frac{-z-(a\,z-1)\,y(z)}{z^2},\ y(0)=0\right\},\ \text{"implies"},\ y(z)$$

$$=\cfrac{z}{1+\cfrac{a_1\,z}{1+\cfrac{\dots}{1+\cfrac{a_n\,z}{1+\dots}}}},\ a_n=\begin{cases}\frac{1}{2}\,n & n::even\\ -a+\frac{1}{2}+\frac{1}{2}\,n & n::odd\end{cases}$$

"12.6.21"

$$-\frac{1}{\left(\frac{1}{z}\right)^l\left(\frac{1}{z}\right)^a\mathrm{e}^{-\frac{1}{z}}\left(\frac{1}{z}\right)^{l+a}\Gamma(a+1+l)}\Bigg(\mathrm{pochhammer}(a,l)\Bigg(-\mathrm{e}^{-\frac{1}{z}}\left(\frac{1}{z}\right)^a\left(\frac{1}{z}\right)^l\Gamma\left(l+a,\frac{1}{z}\right)\Gamma(a)\,\mathrm{e}^{\frac{1}{z}}\,a-\mathrm{e}^{-\frac{1}{z}}\left(\frac{1}{z}\right)^a\left(\frac{1}{z}\right)^l\Gamma\left(l+a,\frac{1}{z}\right)\Gamma(a)\,\mathrm{e}^{\frac{1}{z}}\,l+\mathrm{e}^{-\frac{1}{z}}\left(\frac{1}{z}\right)^a\left(\frac{1}{z}\right)^l\Gamma(a+1+l)\,\Gamma(a)\,\mathrm{e}^{\frac{1}{z}}+\mathrm{e}^{-\frac{1}{z}}\left(\frac{1}{z}\right)^{l+a}\Gamma(a+1+l)\,\Gamma\left(a,\frac{1}{z}\right)\mathrm{e}^{\frac{1}{z}}-\mathrm{e}^{-\frac{1}{z}}\left(\frac{1}{z}\right)^{l+a}\Gamma(a+1+l)\,\Gamma(a)\,\mathrm{e}^{\frac{1}{z}}-\left(\frac{1}{z}\right)^{l+a}\Gamma(a+1+l)\,\Gamma\left(a,\frac{1}{z}\right)\Bigg)\Bigg),\ l=3$$

$$\left\{\frac{d}{dz}y(z)+\frac{-z-(a\,z+3\,z-1)\,y(z)}{z^2},\ y(0)=0\right\},\ \text{"implies"},\ y(z)$$

$$=\cfrac{z}{1+\cfrac{a_1\,z}{1+\cfrac{\dots}{1+\cfrac{a_n\,z}{1+\dots}}}},\ a_n=\begin{cases}\frac{1}{2}\,n & n::even\\ -a-\frac{5}{2}+\frac{1}{2}\,n & n::odd\end{cases}$$

"12.6.22"

$$-\Bigg(\bigg(\left(-\frac{1}{z}\right) l \left(\frac{\left(\frac{1}{z}\right)^{a} e^{-\frac{1}{z}} \operatorname{pochhammer}(-a+1, l) \operatorname{hypergeom}([1, 1-a+l], [\ ], -z)}{z} + \left(\frac{1}{z}\right)^{a} e^{-\frac{1}{z}} \left(-\frac{1}{z}\right)^{l+1} \operatorname{hypergeom}([1, -a+1], [\ ], -z) - \frac{\Gamma\left(a, \frac{1}{z}\right) \left(-\frac{1}{z}\right)^{l+1}}{z}\right) z\bigg) \Big/ \left(\operatorname{pochhammer}(-a+1, l) \left(\frac{1}{z}\right)^{a} e^{-\frac{1}{z}} \left(-\frac{1}{z}\right)^{l+1}\right), l=3$$

$$\left\{\frac{d}{dz} y(z) + \frac{-z-(a z-3 z-1) y(z)}{z^2}, y(0)=0\right\}, \text{"implies"}, y(z) = \frac{z}{1+\frac{a_1 z}{1+\frac{\dots}{1+\frac{a_n z}{1+\dots}}}}, a_n = \begin{cases} \frac{1}{2} n & n::even \\ -a+\frac{7}{2}+\frac{1}{2} n & n::odd \end{cases}$$

"12.6.23"

$$\left\{\frac{d}{dz} y(z) + \frac{-z-(1+z-a) y(z)}{z}, y(0)=0\right\}, \text{"implies"}, y(z) = \frac{z}{a\left(1+\frac{a_1 z}{1+\frac{\dots}{1+\frac{a_n z}{1+\dots}}}\right)}, a_n = \begin{cases} \frac{1}{2} \frac{n}{(a+n)(a+n-1)} & n::even \\ -\frac{1}{2} \frac{2a+n-1}{(a+n)(a+n-1)} & n::odd \end{cases} \quad \textbf{(2.2.4)}$$

Chapter 13

In this chapter, one parameter (k) is given a value, and some variable changes are performed for simplicity.

```
> chapter:=13;
```

$$chapter := 13 \quad (2.3.1)$$

```
> labels(chapter);
```

$$[1.11, 2.20, 3.5, 4.9] \quad (2.3.2)$$

```
> valk := 3;
```

$$valk := 3 \quad (2.3.3)$$

```
> N:=34:
> for formula in labels(chapter) do

    print( sprintf("%a.%a",chapter,formula) );

    f := Cuyt_Cfrac[chapter][formula];

    if formula = 3.5 then   f := subs(k=valk,f);
    elif formula = 4.9 then  f := subs(z=z/sqrt(I*Pi),f)
    end if;

    print(f);

    S := MultiSeries:-series(f,z,N);

    Riccati_to_Cfrac( S, y, z )

  od:
```

"13.1.11"

$$\sqrt{\pi}\, z\, e^{z^2} \operatorname{erf}(z)$$

$$\left\{\frac{d}{dz} y(z) + \frac{-2 z^2 - (2 z^2 + 1) y(z)}{z}, y(0) = 0\right\}, \text{"implies"}, y(z)$$

$$= \frac{2 z^2}{1 + \frac{a_1 z^2}{1 + \frac{\dots}{1 + \frac{a_n z^2}{1 + \dots}}}}, a_n = \begin{cases} \frac{2 n}{(-1 + 2 n) (2 n + 1)} & n::even \\ -\frac{2 n}{(-1 + 2 n) (2 n + 1)} & n::odd \end{cases}$$

"13.2.20"

$$\sqrt{\pi} \operatorname{erfc}\left(\frac{1}{z}\right) e^{\frac{1}{z^2}} z$$

$$\left\{\frac{d}{dz}y(z)+\frac{-2z^2-(z^2-2)y(z)}{z^3}, y(0)=0\right\}, \text{"implies"}, y(z) = \cfrac{z^2}{1+\cfrac{a_1 z^2}{1+\cfrac{\dots}{1+\cfrac{a_n z^2}{1+\dots}}}}, a_n=\frac{1}{2}n$$

"13.3.5"

$$-\frac{1}{3}\left(3\sqrt{\pi}\,\mathrm{erf}\left(\frac{1}{z}\right)e^{\frac{1}{z^2}}z^2-3\sqrt{\pi}\,e^{\frac{1}{z^2}}z^2+2\sqrt{\pi}\,\mathrm{erf}\left(\frac{1}{z}\right)e^{\frac{1}{z^2}}+2z^3-2\sqrt{\pi}\,e^{\frac{1}{z^2}}+2z\right)\Big/\left(z\left(\sqrt{\pi}\,\mathrm{erf}\left(\frac{1}{z}\right)e^{\frac{1}{z^2}}z^2-\sqrt{\pi}\,e^{\frac{1}{z^2}}z^2+2\sqrt{\pi}\,\mathrm{erf}\left(\frac{1}{z}\right)e^{\frac{1}{z^2}}-2\sqrt{\pi}\,e^{\frac{1}{z^2}}+2z\right)\right)$$

$$\left\{\frac{d}{dz}y(z)-\frac{z-2y(z)-6zy(z)^2}{z^3}, y(0)=0\right\}, \text{"implies"}, y(z) = \frac{1}{2}\cfrac{z}{1+\cfrac{a_1 z^2}{1+\cfrac{\dots}{1+\cfrac{a_n z^2}{1+\dots}}}}, a_n=\frac{1}{2}n+\frac{3}{2}$$

"13.4.9"

$$\frac{\left(\mathrm{FresnelC}\left(\frac{z}{\sqrt{I\pi}}\right)+I\,\mathrm{FresnelS}\left(\frac{z}{\sqrt{I\pi}}\right)\right)e^{-\frac{1}{2}z^2}\sqrt{I\pi}}{z}$$

$$\left\{\frac{d}{dz}y(z)-\frac{1-(z^2+1)y(z)}{z}, y(0)=1\right\}, \text{"implies"}, y(z)=1 - \frac{1}{3}\cfrac{z^2}{1+\cfrac{a_1 z^2}{1+\cfrac{\dots}{1+\cfrac{a_n z^2}{1+\dots}}}}, a_n=\begin{cases}-\frac{n}{(2n+3)(2n+1)} & n::even\\ \frac{n+2}{(2n+3)(2n+1)} & n::odd\end{cases} \tag{2.3.4}$$

Chapter 14

```
> chapter := 14;
```

$$chapter := 14 \quad \textbf{(2.4.1)}$$

```
> N:=20;
```

$$N := 20 \quad \textbf{(2.4.2)}$$

```
> labels(chapter);
```

$$[1.16, 1.20, 2.24] \quad \textbf{(2.4.3)}$$

Parameters n and nu are given example values.

```
> valn := 4;
  valnu := 3/2;
```

$$valn := 4$$

$$valnu := \frac{3}{2} \quad \textbf{(2.4.4)}$$

```
> for formula in labels(chapter) do

    print( sprintf("%a.%a",chapter,formula) );

    f := Cuyt_Cfrac[chapter][formula];
    print(f);

    if formula = 1.16 then
      S := MultiSeries:-series(f,z,N) assuming nu::real;
    elif formula = 1.20 then
      S := MultiSeries:-series(f,z,N) assuming 1-nu > 0;
    else
      S := MultiSeries:-series(f,z,N);
    fi;

    Riccati_to_Cfrac( S, y, z )

  od:
```

"14.1.16"

$$e^{\frac{1}{z}} \operatorname{Ei}\left(\nu, \frac{1}{z}\right)$$

$$\left\{\frac{d}{dz} y(z) - \frac{z - (\nu z - z + 1)\, y(z)}{z^2},\ y(0) = 0\right\}, \text{"implies"}, y(z)$$

$$= \cfrac{z}{1 + \cfrac{a_1 z}{1 + \cfrac{\dots}{1 + \cfrac{a_n z}{1 + \dots}}}},\ a_n = \begin{cases} \frac{1}{2} n & n::even \\ \nu - \frac{1}{2} + \frac{1}{2} n & n::odd \end{cases}$$

"14.1.20"

$$-\frac{z\left(-z^{\nu-1}\Gamma(1-\nu)+\mathrm{Ei}(\nu,z)\right)}{e^{-z}}$$

$$\left\{\frac{d}{dz}y(z)+\frac{-z-(z+\nu)\,y(z)}{z},\ y(0)=0\right\}, \text{"implies"},\ y(z)=$$

$$-\frac{z}{(\nu-1)\left(1+\cfrac{a_1 z}{1+\cfrac{\dots}{1+\cfrac{a_n z}{1+\dots}}}\right)},\ a_n=$$

$$\begin{cases}\frac{1}{2}\,\frac{n}{(n-\nu)\,(-\nu+n+1)} & n::even\\ -\frac{1}{2}\,\frac{-2\,\nu+n+1}{(n-\nu)\,(-\nu+n+1)} & n::odd\end{cases}$$

"14.2.24"

$$-\frac{\left(\frac{\mathrm{Ei}\left(-\frac{1}{z}\right)e^{\frac{1}{z}}}{z}-1\right)z}{\mathrm{Ei}\left(-\frac{1}{z}\right)e^{\frac{1}{z}}}$$

$$\left\{\frac{d}{dz}y(z)+\frac{-z-2-(z+3)\,y(z)-y(z)^2}{z^2},\ y(0)=-2\right\}, \text{"implies"},\ y(z)=-2 \qquad \textbf{(2.4.5)}$$

$$-\frac{z}{1+\cfrac{a_1 z}{1+\cfrac{\dots}{1+\cfrac{a_n z}{1+\dots}}}},\ a_n=\begin{cases}1+\frac{1}{2}\,n & n::even\\ \frac{1}{2}\,n+\frac{1}{2} & n::odd\end{cases}$$

Chapter 15

Chapter 15 contains parametrized hypergeometric ratios, which are long to compute with.
For this reason, information is printed along the computation, and additional arguments are used, as detailed below.

```
> chapter := 15;
```

$$chapter := 15 \qquad \textbf{(2.5.1)}$$

```
> N:=20;
```

$$N := 20 \qquad (2.5.2)$$

```
> labels(chapter);
```

$$[3.3,\ 3.4] \qquad (2.5.3)$$

```
> infolevel[demo]:=1;
```

$$infolevel_{demo} := 1 \qquad (2.5.4)$$

```
> for formula in labels(chapter) do
    print( sprintf("%a.%a",chapter,formula) );
    f := Cuyt_Cfrac[chapter][formula];
    print(f);
    S := series(f,z,N);
    Riccati_to_Cfrac( S, y, z, 15 );
  od:
```

"15.3.3"

$$\frac{\mathrm{hypergeom}([a, b], [c], z)}{\mathrm{hypergeom}([a, b+1], [c+1], z)}$$

```
computing series...
... done.
computing a Riccati equation (using a guessing approach)...
... done.
computing a C-fraction expansion...
... done. (in 106.252 seconds)
```

$$\left\{ \frac{\mathrm{d}}{\mathrm{d}z}\, y(z) + \frac{z\, b\, (a-c) - c\, (a\, z - b\, z - c)\, y(z) - c^2\, y(z)^2}{c\, (z-1)\, z},\ y(0) = 1 \right\},$$

$$\text{"implies"},\ y(z) = 1 + \cfrac{a_0\, z}{1 + \cfrac{\dots}{1 + \cfrac{a_n\, z}{1 + \dots}}},\ a_n =$$

$$\begin{cases} \frac{1}{4}\, \frac{(2\, b - 2\, c - n)\, (2\, a + n)}{(c+n)\, (1 + c + n)} & n::even \\ \frac{1}{4}\, \frac{(2\, b + n + 1)\, (2\, a - 2\, c - n - 1)}{(c+n)\, (1 + c + n)} & n::odd \end{cases}$$

"15.3.4"

$$z\, \mathrm{hypergeom}([1, a], [c+1], z)$$

```
computing series...
... done.
computing a Riccati equation (using a guessing approach)...
```

```
... done.
computing a C-fraction expansion...
... done. (in 7.454 seconds)
```

$$\left\{\frac{d}{dz}\,y(z) - \frac{-c\,z-(a\,z-c-z+1)\,y(z)}{z\,(z-1)},\ y(0)=0\right\}, \text{"implies"}, y(z) = \cfrac{z}{1+\cfrac{a_1\,z}{1+\cfrac{...}{1+\cfrac{a_n\,z}{1+...}}}}, a_n = \begin{cases} \frac{1}{4}\,\frac{n\,(2\,a-2\,c-n)}{(c+n)\,(c+n-1)} & n{::}even \\ -\frac{1}{4}\,\frac{(2\,c+n-1)\,(2\,a+n-1)}{(c+n)\,(c+n-1)} & n{::}odd \end{cases} \quad \textbf{(2.5.5)}$$

```
> infolevel[demo]:=0;
```

$$infolevel_{demo} := 0 \quad \textbf{(2.5.6)}$$

Chapter 16

```
> chapter := 16;
```

$$chapter := 16 \quad \textbf{(2.6.1)}$$

```
> N:=20;
```

$$N := 20 \quad \textbf{(2.6.2)}$$

```
> labels(chapter);
```

$$[1.13, 1.14, 2.4, 3.4, 5.7] \quad \textbf{(2.6.3)}$$

```
> for formula in labels(chapter) do
    print( sprintf("%a.%a",chapter,formula) );
    f := Cuyt_Cfrac[chapter][formula];
    print(f);
    S := series(f,z,N);
    Riccati_to_Cfrac( S, y, z );
  od:
```

"16.1.13"

$$\frac{\text{hypergeom}([a],[b],z)}{\text{hypergeom}([a+1],[b+1],z)}$$

$$\left\{\frac{d}{dz}y(z)-\frac{z\,a+b\,(b-z)\,y(z)-b^2\,y(z)^2}{z\,b},\ y(0)=1\right\},\ \text{"implies"},\ y(z)=1$$

$$+\frac{a_0\,z}{1+\frac{\dots}{1+\frac{a_n\,z}{1+\dots}}},\ a_n=\begin{cases}\frac{1}{2}\,\frac{-2\,b-n+2\,a}{(1+n+b)\,(n+b)} & n::even\\ \frac{1}{2}\,\frac{n+1+2\,a}{(1+n+b)\,(n+b)} & n::odd\end{cases}$$

"16.1.14"

$$z\,\mathrm{hypergeom}([1],[b+1],z)$$

$$\left\{\frac{d}{dz}y(z)+\frac{-z\,b-(1+z-b)\,y(z)}{z},\ y(0)=0\right\},\ \text{"implies"},\ y(z)$$

$$=\frac{z}{1+\frac{a_1\,z}{1+\frac{\dots}{1+\frac{a_n\,z}{1+\dots}}}},\ a_n=\begin{cases}\frac{1}{2}\,\frac{n}{(b+n)\,(b+n-1)} & n::even\\ -\frac{1}{2}\,\frac{2\,b+n-1}{(b+n)\,(b+n-1)} & n::odd\end{cases}$$

"16.2.4"

$$\frac{\mathrm{hypergeom}([a,b],[\ ],z)}{\mathrm{hypergeom}([a,b+1],[\ ],z)}$$

$$\left\{\frac{d}{dz}y(z)+\frac{-z\,b-(a\,z-b\,z-1)\,y(z)-y(z)^2}{z^2},\ y(0)=1\right\},\ \text{"implies"},\ y(z)$$

$$=1+\frac{a_0\,z}{1+\frac{\dots}{1+\frac{a_n\,z}{1+\dots}}},\ a_n=\begin{cases}-a-\frac{1}{2}\,n & n::even\\ -b-\frac{1}{2}-\frac{1}{2}\,n & n::odd\end{cases}$$

"16.3.4"

$$\frac{\mathrm{hypergeom}([\ ],[b],z)}{\mathrm{hypergeom}([\ ],[b+1],z)}$$

$$\left\{\frac{d}{dz}y(z)-\frac{z+b^2\,y(z)-b^2\,y(z)^2}{z\,b},\ y(0)=1\right\},\ \text{"implies"},\ y(z)=1$$

$$+\frac{a_0\,z}{1+\frac{\dots}{1+\frac{a_n\,z}{1+\dots}}},\ a_n=\frac{1}{(n+b)\,(n+1+b)}$$

"16.5.7"

$$e^z$$

$$\left\{\frac{d}{dz}y(z)-y(z),\ y(0)=1\right\},\ \text{"implies"},\ y(z)=1+\cfrac{z}{1+\cfrac{a_1 z}{1+\cfrac{\dots}{1+\cfrac{a_n z}{1+\dots}}}},\ a_n \quad \mathbf{(2.6.4)}$$

$$=\begin{cases}\frac{1}{2(n+1)} & n::even\\ -\frac{1}{2n} & n::odd\end{cases}$$

Chapter 17

In this chapter, some functions require to compute the series by hand, as well as the Riccati equation.
The formulas are still produced and proved automatically.

```
> chapter := 17;
```

$$chapter := 17 \quad \mathbf{(2.7.1)}$$

```
> labels(chapter);
```

$$[1.37, 1.39, 1.44, 1.46, 2.32, 2.35] \quad \mathbf{(2.7.2)}$$

```
> N:=20;
```

$$N := 20 \quad \mathbf{(2.7.3)}$$

```
> for formula in labels(chapter) do

    print( sprintf("%a.%a",chapter,formula) );

    f := subs(n=nn, Cuyt_Cfrac[chapter][formula]);



    if formula = 1.44 then

      lprint("Ad-hoc series computation, and Riccati equation 
  computation.");
         f := subs(z=z/I,f/exp(-I*Pi));
       print(f);
        f := eval(f, HankelH1 = ContFrac:-myHankelH1);
        f := simplify(f,power,symbolic,exp);
        S := map(simplify,MultiSeries:-series( f, z, N ));

        deq := diff(y(z), z) - (z^2+(2*nu+2)*z^2*y(z)-y(z)^2)
  /z^3:
        Riccati_to_Cfrac(  S, y, z, input_riccati = deq );
```

```
    elif formula = 1.46 then

      lprint("Ad-hoc series computation.");
        f := subs(z=I*z,f);
       print(f);
        f := eval(f, HankelH2 = ContFrac:-myHankelH2);
        f := simplify(f,power,symbolic,exp);
        S := map(simplify,MultiSeries:-series( f, z, N ));

        Riccati_to_Cfrac(  S, y, z );

    elif formula = 2.35 then

      print(f);
      S := map(factor,series(f,z,N));
      Riccati_to_Cfrac(  S, y, z );

    else

       print(f);
      S := MultiSeries:-series( f, z, N );
      Riccati_to_Cfrac(  S, y, z );

     fi;

  od:
```

"17.1.37"

$$\frac{\mathrm{BesselJ}(\nu+1, z)}{\mathrm{BesselJ}(\nu, z)}$$

$$\left\{\frac{d}{dz}\, y(z) + \frac{-z - (-1-2\nu)\, y(z) - z\, y(z)^2}{z},\ y(0)=0\right\}, \text{"implies"}, y(z) = \frac{1}{2}\, \frac{z}{(\nu+1)\left(1 + \cfrac{a_1 z^2}{1 + \cfrac{\dots}{1 + \cfrac{a_n z^2}{1+\dots}}}\right)},\ a_n = -\frac{1}{4\,(n+\nu)\,(\nu+n+1)}$$

"17.1.39"

$$-\frac{1}{2}\, \frac{2\,\mathrm{BesselI}(nn+1, z)\,\nu - z\,\mathrm{BesselI}(nn, z) + 2\,\mathrm{BesselI}(nn+1, z)}{\mathrm{BesselI}(nn+1, z)\,(\nu+1)}$$

$$\left\{\frac{d}{dz}\, y(z) - \frac{1}{2}\, \frac{1}{z\,(\nu+1)}\left(4\,\nu\, nn - 4\,\nu^2 + z^2 + 4\, nn - 4\,\nu + 4\,(\nu+1)\,(-2\,\nu + nn - 1)\, y(z) - 4\,(\nu+1)^2\, y(z)^2\right),\ y(0) = \frac{nn-\nu}{\nu+1}\right\}, \text{"implies"}, y(z)$$

$$= \frac{nn - \nu}{\nu + 1} + \frac{1}{4} \frac{z^2}{(nn\,\nu + nn + 2\,\nu + 2)\left(1 + \cfrac{a_1 z^2}{1 + \cfrac{\dots}{1 + \cfrac{a_n z^2}{1 + \dots}}}\right)}, a_n$$

$$= \frac{1}{4\,(nn + n + 1)\,(nn + n + 2)}$$

"17.1.44"

"Ad-hoc series computation, and Riccati equation computation."

$$\frac{\mathrm{I}\, z\, \mathrm{HankelH1}\left(\nu + 1, \frac{\mathrm{I}}{z}\right)}{\mathrm{HankelH1}\left(\nu, \frac{\mathrm{I}}{z}\right)}$$

$$\left\{\frac{\mathrm{d}}{\mathrm{d}z} y(z) - \frac{z^2 + (2\,\nu + 2)\, z^2\, y(z) - y(z)^2}{z^3}, y(0) = 0\right\}, \text{"implies"}, y(z)$$

$$= \frac{z}{1 + \cfrac{-\nu\, z - \frac{1}{2}\, z}{1 + \cfrac{a_2\, z}{1 + \cfrac{\dots}{1 + \cfrac{a_n\, z}{1 + \dots}}}}}, a_n = \begin{cases} \frac{1}{2}\,\nu + \frac{1}{4} + \frac{1}{4}\, n & n{::}even \\ -\frac{1}{2}\,\nu - \frac{1}{2} + \frac{1}{4}\, n & n{::}odd \end{cases}$$

"17.1.46"

"Ad-hoc series computation."

$$\frac{\mathrm{I}\, z\, \mathrm{HankelH2}\left(\nu + 1, -\frac{\mathrm{I}}{z}\right)}{\mathrm{HankelH2}\left(\nu, -\frac{\mathrm{I}}{z}\right)}$$

$$\left\{\frac{\mathrm{d}}{\mathrm{d}z} y(z) + \frac{z^2 - (2\,\nu\, z^2 + 2\, z^2)\, y(z) - y(z)^2}{z^3}, y(0) = 0\right\}, \text{"implies"}, y(z) =$$

$$-\frac{z}{1+\frac{-\nu z-\frac{1}{2}z}{1+\frac{a_2 z}{1+\frac{\ldots}{1+\frac{a_n z}{1+\ldots}}}}},\ a_n = \begin{cases} \frac{1}{2}\nu+\frac{1}{4}+\frac{1}{4}n & n::even \\ -\frac{1}{2}\nu-\frac{1}{2}+\frac{1}{4}n & n::odd \end{cases}$$

"17.2.32"

$$-\frac{1}{2}\,\frac{2\,\mathrm{BesselI}(\nu+1,z)\,\nu - z\,\mathrm{BesselI}(\nu,z)+2\,\mathrm{BesselI}(\nu+1,z)}{\mathrm{BesselI}(\nu+1,z)\,(\nu+1)}$$

$$\left\{\frac{d}{dz}y(z)-\frac{1}{2}\,\frac{z^2-(4\nu^2+8\nu+4)\,y(z)-(4\nu^2+8\nu+4)\,y(z)^2}{z(\nu+1)},\ y(0)=0\right\},\ \text{"implies"},\ y(z)=\frac{1}{4}\,\frac{z^2}{(\nu^2+3\nu+2)\left(1+\frac{a_1 z^2}{1+\frac{\ldots}{1+\frac{a_n z^2}{1+\ldots}}}\right)},\ a_n$$

$$=\frac{1}{4(\nu+n+1)(\nu+n+2)}$$

"17.2.35"

$$\frac{1}{2}\left(\left(2\,\mathrm{BesselK}\left(\nu+1,\frac{1}{z}\right)\nu-\frac{2\,\mathrm{BesselK}\left(\nu+1,\frac{1}{z}\right)}{z}+\frac{2\,\mathrm{BesselK}\left(\nu,\frac{1}{z}\right)}{z}+\mathrm{BesselK}\left(\nu+1,\frac{1}{z}\right)\right)z\right)\Big/\left(-\mathrm{BesselK}\left(\nu,\frac{1}{z}\right)+\mathrm{BesselK}\left(\nu+1,\frac{1}{z}\right)\right)$$

$$\left\{\frac{d}{dz}y(z)-\frac{1}{2}\,\frac{2\nu z+3z-(-4\nu z-4z+4)\,y(z)-4\,y(z)^2}{z^2},\ y(0)=0\right\}, \tag{2.7.4}$$

$$\text{"implies"},\ y(z)=\frac{a_0 z}{1+\frac{\ldots}{1+\frac{a_n z}{1+\ldots}}},\ a_n=\begin{cases}\frac{1}{2}\nu+\frac{3}{4}+\frac{1}{4}n & n::even\\ -\frac{1}{2}\nu+\frac{1}{4}n & n::odd\end{cases}$$

Chapter 18

```
> chapter := 18;
```

$$chapter := 18 \quad (2.8.1)$$

```
> labels(chapter);
```

$$[2.16, 2.17, 2.22, 3.4, 4.20] \quad (2.8.2)$$

```
> N:=40;
```

$$N := 40 \quad (2.8.3)$$

```
> alpha := 2;
```

$$\alpha := 2 \quad (2.8.4)$$

```
> for formula in labels(chapter) do

    print( sprintf("%a.%a",chapter,formula) );

     f := Cuyt_Cfrac[chapter][formula];
    print(f);
     S := MultiSeries:-series( f, z, N );
     Riccati_to_Cfrac(  S, y, z );

  od:
```

"18.2.16"

$$\frac{1}{2} \frac{\operatorname{erfc}\left(\frac{1}{2} \frac{\sqrt{2}}{z}\right) z \sqrt{2} \sqrt{\pi}}{e^{-\frac{1}{2 z^2}}}$$

$$\left\{\frac{d}{dz} y(z) + \frac{-z^2 - (z^2 - 1) y(z)}{z^3}, y(0) = 0\right\}, \text{"implies"}, y(z)$$

$$= \cfrac{z^2}{1 + \cfrac{a_1 z^2}{1 + \cfrac{\ldots}{1 + \cfrac{a_n z^2}{1 + \ldots}}}}, a_n = n$$

"18.2.17"

$$-\frac{e^{-\frac{1}{2} z^2} z \sqrt{2} + \operatorname{erfc}\left(\frac{1}{2} z \sqrt{2}\right) \sqrt{\pi} - \sqrt{\pi}}{\sqrt{\pi} \left(\operatorname{erfc}\left(\frac{1}{2} z \sqrt{2}\right) - 1\right)}$$

$$\left\{\frac{d}{dz} y(z) - \frac{-z^2 - (z^2 + 1) y(z) - y(z)^2}{z}, y(0) = 0\right\}, \text{"implies"}, y(z)$$

$$= \frac{a_0 z^2}{1 + \frac{\ldots}{1 + \frac{a_n z^2}{1 + \ldots}}},\ a_n = \begin{cases} -\frac{n+1}{(2n+1)(2n+3)} & n::even \\ \frac{n+1}{(2n+1)(2n+3)} & n::odd \end{cases}$$

"18.2.22"

$$-\frac{1}{2}\,\frac{\left(\sqrt{2}\,\sqrt{\pi}\,e^{\frac{1}{2}z^2}\,\mathrm{erfc}\left(\frac{1}{2}\,z\sqrt{2}\right) - \sqrt{2}\,\sqrt{\pi}\,e^{\frac{1}{2}z^2} + 2z\right)\sqrt{2}}{\sqrt{\pi}\,e^{\frac{1}{2}z^2}\left(\mathrm{erfc}\left(\frac{1}{2}\,z\sqrt{2}\right) - 1\right)}$$

$$\left\{\frac{d}{dz}\,y(z) - \frac{-z^2 - (z^2+1)\,y(z) - y(z)^2}{z},\ y(0) = 0\right\},\ \text{"implies"},\ y(z)$$

$$= \frac{a_0 z^2}{1 + \frac{\ldots}{1 + \frac{a_n z^2}{1 + \ldots}}},\ a_n = \begin{cases} -\frac{n+1}{(2n+1)(2n+3)} & n::even \\ \frac{n+1}{(2n+1)(2n+3)} & n::odd \end{cases}$$

"18.3.4"

$$-\frac{1}{4}\left(\frac{\sqrt{\pi}\,\mathrm{erf}\left(\frac{1}{2}\,\frac{\sqrt{2}}{z}\right)}{z^4} + \frac{\sqrt{2}\,e^{-\frac{1}{2z^2}}}{z^3} - \frac{\sqrt{\pi}}{z^4} + \frac{6\sqrt{\pi}\,\mathrm{erf}\left(\frac{1}{2}\,\frac{\sqrt{2}}{z}\right)}{z^2} + \frac{5\,e^{-\frac{1}{2z^2}}\sqrt{2}}{z} - \frac{6\sqrt{\pi}}{z^2} + 3\sqrt{\pi}\,\mathrm{erf}\left(\frac{1}{2}\,\frac{\sqrt{2}}{z}\right) - 3\sqrt{\pi}\right) \Bigg/ \left(\frac{\sqrt{\pi}\,\mathrm{erf}\left(\frac{1}{2}\,\frac{\sqrt{2}}{z}\right)}{z^3} + \frac{e^{-\frac{1}{2z^2}}\sqrt{2}}{z^2} - \frac{\sqrt{\pi}}{z^3} + \frac{3\sqrt{\pi}\,\mathrm{erf}\left(\frac{1}{2}\,\frac{\sqrt{2}}{z}\right)}{z} + 2\,e^{-\frac{1}{2z^2}}\sqrt{2} - \frac{3\sqrt{\pi}}{z}\right)$$

$$\left\{\frac{d}{dz}\,y(z) - \frac{z - y(z) - 4z\,y(z)^2}{z^3},\ y(0) = 0\right\},\ \text{"implies"},\ y(z)$$

$$= \frac{z}{1+\frac{a_1 z^2}{1+\frac{\dots}{1+\frac{a_n z^2}{1+\dots}}}}, \; a_n = n+4$$

"18.4.20"

$$\frac{\left(1-e^{-\frac{z}{\theta}}\left(1+\frac{z}{\theta}\right)\right)\theta}{z\,e^{-\frac{z}{\theta}}}$$

$$\left\{\frac{d}{dz}\,y(z) + \frac{-z-(z-\theta)\,y(z)}{z\,\theta},\; y(0)=0\right\}, \text{"implies"}, y(z) \tag{2.8.5}$$

$$= \frac{1}{2}\,\frac{z}{\theta\left(1+\frac{a_1 z}{1+\frac{\dots}{1+\frac{a_n z}{1+\dots}}}\right)}, \; a_n =$$

$$\begin{cases} \frac{1}{2}\,\frac{n}{\theta\,(n+2)\,(n+1)} & n::even \\ -\frac{1}{2}\,\frac{n+3}{\theta\,(n+2)\,(n+1)} & n::odd \end{cases}$$

Other examples.

logarithmic derivative of Airy function, at infinity

An indication that this function is of interest appears in the Chudnovsky brothers article of 1991.

We perform a slight linear fractional transformation to remove the first terms, and prove the formula.

```
> f:=AiryAi(x);
```

$$f := \mathrm{AiryAi}(x) \tag{3.1.1}$$

```
> S:=MultiSeries:-asympt(f,x,100):
> dS:=diff(S,x):
> dlog:=asympt(dS/S,x,100):
> St0 := subs(x=1/t^2,dlog):
> series( t*St0 ,t,90):
> St := eval(%,csgn=1):
```

Formula

```
> numtheory[cfrac](St,t,10,simregular):
> numtheory[cfrac](%,t,simregular,quotients);
```

$$\left[-1, \left[-\frac{1}{4}t^3, 1\right], \left[\frac{5}{8}t^3, 1\right], \left[\frac{7}{8}t^3, 1\right], \left[\frac{11}{8}t^3, 1\right], \left[\frac{13}{8}t^3, 1\right], \left[\frac{17}{8}t^3, 1\right], \left[\frac{19}{8}t^3, 1\right], \left[\frac{23}{8}t^3, 1\right], \left[\frac{25}{8}t^3, 1\right], \left[\frac{29}{8}t^3, 1\right], \ldots\right] \quad (3.1.2)$$

Ad hoc transformation to remove the first terms.

```
> S:=eval( series(1+t^3/4/(1+St),t,90), t=z^(1/3)):
> gfun:-ContFrac:-Riccati_to_Cfrac(subs(t=z,S),y,z,10);
```

$$\left\{\frac{d}{dz}y(z) + \frac{1}{6}\frac{5z-(4z-8)\,y(z)-8\,y(z)^2}{z^2}, y(0)=0\right\}, \text{"implies"}, y(z) = -\frac{5}{8}\cfrac{z}{1+\cfrac{a_1 z}{1+\cfrac{\ldots}{1+\cfrac{a_n z}{1+\ldots}}}}, a_n = \begin{cases} \frac{5}{8}+\frac{3}{8}n & n{::}even \\ \frac{1}{2}+\frac{3}{8}n & n{::}odd \end{cases} \quad (3.1.3)$$

dlmf 4.25.5

```
> f:= exp(2*A*arctan(1/z));
```

$$f := e^{2A\arctan\left(\frac{1}{z}\right)} \quad (3.2.1)$$

```
> A:=2/3;
```

$$A := \frac{2}{3} \quad (3.2.2)$$

```
> g:='g';
```

$$g := g \quad (3.2.3)$$

```
> g:=subs(z=1/z, solve(f=1+2*A/(z-A+g(z)),g(z)));
```

$$g := -\frac{1}{3}\frac{\frac{3\,e^{\frac{4}{3}\arctan(z)}}{z} - 2\,e^{\frac{4}{3}\arctan(z)} - \frac{3}{z} - 2}{e^{\frac{4}{3}\arctan(z)} - 1} \quad (3.2.4)$$

```
> G := map(normal,series(g,z,40)):
> gfun:-ContFrac:-Riccati_to_Cfrac(G,y,z,20);
```

$$\left\{\frac{d}{dz}y(z) - \frac{13z-18\,y(z)-9z\,y(z)^2}{9z^3+9z}, y(0)=0\right\}, \text{"implies"}, y(z) \quad (3.2.5)$$

$$= \frac{13}{27} \frac{z}{1 + \frac{a_1 z^2}{1 + \frac{...}{1 + \frac{a_n z^2}{1 + ...}}}}, a_n = -\frac{1}{9} \frac{(-3\,n - 3 + 2\,I)\,(3\,n + 3 + 2\,I)}{(2\,n + 1)\,(2\,n + 3)}$$

>

```
> time()-starttime;
```

193.940 **(2)**